
LET, LAMBDA & co 

Excelstammtisch vom 11.01.2022 

1. Neue Formeln zum Abkürzen 

Bis Excel 2003 musste man Fehler, beispielsweise bei einem SVERWEIS  

=SVERWEIS(D2;A2:B13;2;FALSCH) 

folgendermaßen abfangen: 

=WENN(ISTFEHLER(SVERWEIS(D2;A2:B13;2;FALSCH));""; 
SVERWEIS(D2;A2:B13;2;FALSCH)) 

Dadurch wird die eigentliche Funktion zwei Mal verwendet, was bei langen Funktionen sehr 
mühsam sein konnte. 

In Excel 2007 wurde WENNFEHLER eingeführt: 

=WENNFEHLER(SVERWEIS(D2;A2:B13;2;FALSCH);"") 

Das kürzte die Länge der Funktionen enorm ab. 

2. LET 

Die Funktion LET, die in ihrer Anfangszeit einige andere Namen (beispielsweise SEI) erhalten 
hat, bestreitet den gleichen Weg: 

Funktionen, die mehrmals verwendet werden, können ausgelagert werden. Ein Beispiel: 

=ZELLE("dateiname") 

Liefert den Dateinamen. Beispielsweise 

D:\Eigene 
Dateien\Excel\Excelstammtisch\Excelgimmick20_LET_LAMDA_co\ 
Fertig\[LET01.xlsx]WENNFEHLER 

Steht diese Funktion in A1 kann mit  

=SUCHEN("[";A1) 

Das „[“ gefunden werden. Der Pfad wird herausgelöst mit: 

=LINKS(A1;A2-1) 

Oder in einer Formel: 

=LINKS(ZELLE("dateiname");SUCHEN("[";ZELLE("dateiname"))-1) 

Darin wird allerdings zwei Mal die Funktion  

ZELLE("dateiname") 

Verwendet. LET hilft nun dabei sie unter einem (Variablen-)namen zu speichern (hier: DN): 

=LET(DN;ZELLE("dateiname"); … 

Dann kann dieser Name verwendet werden: 

=LET(DN;ZELLE("dateiname");LINKS(DN;SUCHEN("[";DN)-1)) 

Hinweis 

Der Name der Variablen (hier: DN) darf nicht in Anführungszeichen stehen! 

2.1. Beispiel I für LET: ISBN 

Der Nummernraum der ISBN besteht heute aus 13 Ziffern; früher war er 10-stellig; die zehnte 
Ziffer war die Prüfziffer ist. Jedoch gab es in Osteuropa und im englischen Sprachraum 
Schwierigkeiten, für neue Verlage und Publikationen Nummern zu finden. Anfang 2005 wurde 
die erweiterte 13-stellige ISBN (EAN-13, International Article Number, früher: European Article 
Number) eingeführt. Im Impressum stehen die ISBN-10 und ISBN-13 gleichberechtigt 
nebeneinander. Ab 1. Januar 2007 ist die Angabe der ISBN-13 Pflicht und die ISBN-10 könnte 
entfallen. Normalerweise werden jedoch beide Nummern genannt.  

Die Prüfziffer der „alten“ ISBN-10-Zahl wird folgendermaßen berechnet. 

Die ISBN-10 eines Buches lautet: 3-7460-6487-2. Dort werden die ersten neun Ziffern mit 1, mit 
2, mit 3, … mit 9 multipliziert, also: 

3  1 + 7  2 + 4  3 + 6  4 + 0  6 + 6  4 + 4  7 + 8  8 + 7  9 

Das Ergebnis: 244. Von dieser Zahl wird mod 11 berechnet – also REST(244;11). Das Ergebnis 
ist 2. Dies ist die Prüfziffer. Wäre das Ergebnis 10 (wie beispielsweise in 3-7392-3167-X), dann 
wird ein „X“ verwendet – das römische Zeichen für 10. Steht diese ISBN in der Zelle A1, dann 
könnte beispielsweise die folgende Formel die Prüfziffer berechnen: 

=WENN(REST(TEIL(A1;1;1)*1+TEIL(A1;2;1)*2+TEIL(A1;3;1)*3+ 
TEIL(A1;4;1)*4+TEIL(A1;5;1)*5+TEIL(A1;6;1)*6+TEIL(A1;7;1)*7+ 
TEIL(A1;8;1)*8+TEIL(A1;9;1)*9;11)= 10;"X"; 
REST(TEIL(A1;1;1)*1+TEIL(A1;2;1)*2+TEIL(A1;3;1)*3+ 
TEIL(A1;4;1)*4+TEIL(A1;5;1)*5+TEIL(A1;6;1)*6+ 
TEIL(A1;7;1)*7+TEIL(A1;8;1)*8+TEIL(A1;9;1)*9;11)) 

Prüfen Sie selbst: bei 374318254 lautet die Prüfziffer 8 und bei 373921742 und 373922984 sind es 
1 beziehungsweise 5. 

Man könnte auch diese Prüfziffernberechnung mit einer Matrixfunktion lösen: 

{=WENN(REST(SUMME(TEIL(A1;ZEILE(BEREICH.VERSCHIEBEN 
(B1;0;0;9;1));1)*ZEILE(BEREICH.VERSCHIEBEN 
(B1;0;0;9;1)));11)=10;"X";REST(SUMME(TEIL(A1;ZEILE 
(BEREICH.VERSCHIEBEN(B1;0;0;9;1));1)*ZEILE( 
BEREICH.VERSCHIEBEN(B1;0;0;9;1)));11))} 

Oder mit der neuen Arrayfunktion SEQUENZ: 

=WENN(REST(SUMME(TEIL(A1;SEQUENZ(9);1) 
*SEQUENZ(9));11)=10;"x"; 
REST(SUMME(TEIL(A1;SEQUENZ(9);1)*SEQUENZ(9));11)) 

Hinweis 

Auch in anderen Bereichen finden sich Prüfziffern, beispielsweise bei Kontonummern. 
Da jedoch Banken verschiedene Systeme zur Berechnung der Kontonummern haben, 
würde es den Umfang dieses Kapitels sprengen, sämtliche Varianten hier aufzuzählen, zu 
beschreiben und „nachzurechnen“. 

Allerdings: egal auf welche Art man die ISBN berechnet – man muss zwei Mal TEIL(A1 … 
ermitteln. Diesen kann man auslagern. Im ersten Schritt kann man die Funktion mit LET 
„ummanteln“: 

=LET(T;TEIL(A3;SEQUENZ(9);1); 
WENN(REST(SUMME(TEIL(A3;SEQUENZ(9);1)*SEQUENZ(9));11)=10;"x"; 
REST(SUMME(TEIL(A3;SEQUENZ(9);1)*SEQUENZ(9));11))) 

Dann werden die Teile ersetzt: 

=LET(T;TEIL(A3;SEQUENZ(9);1);WENN(REST(SUMME(T*SEQUENZ(9));11)=10;"x"; 
REST(SUMME(T*SEQUENZ(9));11))) 

Selbstverständlich kann man auch SEQUENZ auslagern: 

=LET(T;TEIL(A3;SEQUENZ(9);1);S;SEQUENZ(9); 
WENN(REST(SUMME(T*S);11)=10;"x";REST(SUMME(T*S);11))) 

Will man im T-Teil die Sequenz(9) durch S ersetzen, muss man S und T vertauschen: 

=LET(S;SEQUENZ(9);T;TEIL(A3;S;1); 
WENN(REST(SUMME(T*S);11)=10;"x";REST(SUMME(T*S);11))) 

Oder noch kürzer: REST(SUMME(T*S);11) wird durch R ersetzt: 

=LET(S;SEQUENZ(9);T;TEIL(A3;S;1);R;REST(SUMME(T*S);11);WENN(R=10;"x";R)) 

2.2. Beispiel II: Schritt für Schritt Teile auslagern: 

Eine Funktion berechnet – ausgehend von einer Jahreszahl, die in der Zelle A1 steht, die 
Kalenderwoche 1, 2, 3, …: 

=TEXT(DATUM($A$1;1;WENN(WOCHENTAG(DATUM($A$1;1;1);2)<=4;1; 
9-WOCHENTAG(DATUM($A$1;1;1);2)))+ 
(ZEILE()-1)*7;"TT.MM.JJJJ")&" - "& 
TEXT(DATUM($A$1;1;WENN(WOCHENTAG(DATUM($A$1;1;1);2)<=4;1; 
9-WOCHENTAG(DATUM($A$1;1;1);2)))+(ZEILE()-1)*7+ 
WENN(WOCHENTAG(DATUM($A$1;1; 
WENN(WOCHENTAG(DATUM($A$1;1;1);2)<=4;1; 
9-WOCHENTAG(DATUM($A$1;1;1);2)))+(ZEILE()-1)*7)=1;7; 
7-WOCHENTAG(DATUM($A$1;1; 
WENN(WOCHENTAG(DATUM($A$1;1;1);2)<=4;1; 
9-WOCHENTAG(DATUM($A$1;1;1);2)))+ 
(ZEILE()-1)*7;2));"TT.MM.JJJJ") 

Nun kann man auf „W“ auslagern: WOCHENTAG(DATUM($A$1;1;1);2) 

Auf WW: DATUM($A$1;1;WENN(W<=4;1;9-W)) 

Auf Z: WW+(ZEILE()-1)*7 

Und schreiben: 

=LET(W;WOCHENTAG(DATUM($A$1;1;1);2); 
WW;DATUM($A$1;1;WENN(W<=4;1;9-W));Z;WW+(ZEILE()-1)*7; 
TEXT(Z;"TT.MM.JJJJ")&" - "& 
TEXT(Z+WENN(WOCHENTAG(Z)=1;7;7-WOCHENTAG(Z;2));"TT.MM.JJJJ")) 

2.3. Beispiel III: Ostern 

Bedauerlicherweise hat Excel keine Funktion zur Berechnung des Ostersonntags. Dies ist deshalb 
bedauerlich, weil eine Reihe von Feiertagen aufgrund des Osterfestes berechnet werden. Auf dem 
Konzil zu Nicäa wurde im Jahre 325 festgelegt, dass Ostersonntag der erste Sonntag nach dem 
ersten Vollmond an oder nach dem Frühlingsanfang sein solle.  

Für den julianischen Kalender ermittelte Gauß die folgende Formel zur Berechnung des 
Ostersonntages im julianischen Jahr J: 

a = J-1900 

b = a mod 19 

c = Abrunden((7  b + 1)/19;0) 

d = 11  b + 4 - c mod 29 

e = Abrunden(a / 4;0) 

f = a + e + 31 -d mod 7 

Ostersonntag ist der 25-d-f.te April. 

 

Das heißt, in Excel könnte man diese Formel (für die Jahre 1999 bis 2099) folgendermaßen 
anwenden: 

=DATUM(B1;4;25-REST(11*REST(B1-1900;19)+4- 
ABRUNDEN((7*REST(B1-1900;19)+1)/19;0);29) 
-REST(B1-1900+ABRUNDEN((B1-1900)/4;0)+31 
-REST(11*REST(B1-1900;19)+4- 
ABRUNDEN((7*REST(B1-1900;19)+1)/19;0);29);7)) 

Oder man lagert die Teile in die Variablen a, b, c, d, e und f aus: 

=LET(a;B1-1900;b;REST(a;19);c;ABRUNDEN((7*b+1)/19;0); 
d;REST(11*b+4-c;29);e;ABRUNDEN((a)/4;0);f;REST(a+e+31-d;7); 
DATUM(B1;4;25-d-f)) 

2.4. Beispiel IV: Zahlen in Texte 

Ein hübsches und beeindruckendes (spanisches Beispiel) habe ich im Internet gefunden: 

=LET(importe;A1; 
QMillones;WENN(importe>999999;GANZZAHL(importe/1000000)); 
QMiles;WENN(importe>999;GANZZAHL(importe/1000)-QMillones*1000); 
QCentenas;importe-GANZZAHL(importe/1000)*1000; 
QEntera;GANZZAHL(importe); 
QDecimal;RUNDEN((importe-GANZZAHL(importe))*100;0); 
 
centfin;WENN(QDecimal=1;" céntimo";" céntimos"); 
eurfin;WENN(QEntera=1;" euro";" euros"); 
 
matriz;WAHL({1;2;3;4};QMillones;QMiles;QCentenas;QDecimal); 
calculo;LET( 
centena2;GANZZAHL(matriz/100); 
decena2;GANZZAHL((matriz-centena2*100)/10); 
unidad2;GANZZAHL((matriz-centena2*100-decena2*10)); 
 
centenas;{""."cien"."doscientos"."trescientos"."cuatrocientos"."quiniento
s"."seiscientos"."setecientos"."ochocientos"."novecientos"}; 
decenas;{""." diez"." veinte"." treinta"." cuarenta"." cincuenta"." 
sesenta"." setenta"." ochenta"." noventa"}; 
unidades;{""." un"." dos"." tres"." cuatro"." cinco"." seis"." siete"." 

ocho"." nueve"}; 
dieces;{"diez"."once"."doce"."trece"."catorce"."quince"."dieciseis"."diec
isiete"."dieciocho"."diecinueve"}; 
veintes;{"veinte"."veintiuno"."veintidos"."veintitres"."veinticuatro"."ve
inticinco"."veintiseis"."veintisiete"."veintiocho"."veintinueve"}; 
tatantos;{""." y un"." y dos"." y tres"." y cuatro"." y cinco"." y 
seis"." y siete"." y ocho"." y nueve"}; 
 
num_letra1;INDEX(centenas;1;centena2+1); 
num_letra2;WENN(centena2=1;num_letra1&"to";num_letra1); 
num_letra3;num_letra2&INDEX(decenas;1;decena2+1); 
num_letra4;WENN(decena2=0;INDEX(unidades;1;unidad2+1);""); 
num_letra5;WENN(decena2=1;TEIL(num_letra3;1;LÄNGE(num_letra3)-
4)&INDEX(dieces;1;unidad2+1);""); 
num_letra6;WENN(decena2=2;TEIL(num_letra3;1;LÄNGE(num_letra3)-
6)&INDEX(veintes;1;unidad2+1);""); 
num_letra7;WENN(decena2>2;INDEX(tatantos;1;unidad2+1);""); 
num_letra8;WENN(decena2=1;num_letra5;WENN(decena2=2;num_letra6;num_letra3
&num_letra4&num_letra7)); 
WENN(matriz=100;"cien";num_letra8)); 
 
txtMillones;WENN(importe>999999;WENN(QMillones=1;"un millón 
";INDEX(calculo;1)&" millones ");""); 
txtMil;WENN(importe>999;WENN(QMiles=1;"mil 
";WENN(QMiles=0;"";INDEX(calculo;2)&" mil "));""); 
txtCent;WENN(QEntera=0;"";WENN(INDEX(calculo;3)="";"";INDEX(calculo;3))&e
urfin); 
txtDecimal;WENN(INDEX(calculo;4)="";"";WENN(QEntera=0;"";" con 
")&INDEX(calculo;4)&centfin); 
 
WENN(importe=0;"cero 
euros";GLÄTTEN(txtMillones&txtMil&txtCent&txtDecimal))) 

Auffällig an diesem Beispiel ist, dass eine der Variablen Bezug auf eine feste Zelle nimmt (A1) 
und dass LET mehrmals ineinander geschachtelt wurde. 

3. LAMBDA 

Einen Schritt weiter geht LAMBDA. Diese Funktion besteht aus mindestens zwei Teilen: 
Variablen und Funktionen. Beispielsweise soll die Differenz aus MAX und MIN berechnet 
werden: 

=LAMBDA(Bereich;MAX(Bereich)-MIN(Bereich)) 

Man kann die Funktion testen: 

=LAMBDA(Bereich;MAX(Bereich)-MIN(Bereich))(A1:A3) 

Und diese Funktion im Namensmanager unter einem Namen speichern: 

Und schließlich verwenden: 

=Spannweite(A1:A3) 

Hinweis 

Zwar wird sie per Intellisense aufgelistet, zwar kann man sie über fx anzeigen lassen, 
aber sie erscheint nicht im Funktionsassistenten. 

Hinweis 

Und: da diese Funktion lokal für diese Datei arbeitet, kann sie leider nicht global für 
Excel, das heißt für mehrere Arbeitsmappen zur Verfügung gestellt werden. Das 
bedeutet: Man kann sie weder als Add-In noch in der Personal.xslb speichern, um sie 
überall zur Verfügung zu stellen. 

Lambda kann mehrere Parameter aufnehmen. 

4. Rekursion in Lambda 

Es gibt drei Möglichkeiten in Excel rekursiv zu rechnen: 

1.) Aktiviert man in den Excel-Optionen in der Kategorie „Formeln“ die iterative Berechnung, 
kann man berechnete Endwerte wieder als Startwerte verwenden, ohne dass es zu einem 
Zirkelbezug kommt. Dort wird die Anzahl der Schritt angeben – das Maximum liegt bei 10.000. 

2.) Erstellt man in VBA eine Funktion, die sich selbst wieder aufruft, kann man diese Funktion 
von einer anderen Funktion aufrufen. Dabei muss ein Abbruchkriterium vorhanden sein! 

3.) Es funktioniert auch mit der Funktion LAMBDA 

Auf der Seite 

https://www.ablebits.com/office-addins-blog/2021/06/16/write-recursive-lambda-function-
excel/ 

beschreibt Svetlana Cheusheva wie man hierbei vorgehen kann: 

In einer Zelle findet sich ein Text 

Rene%&/()%&/()Bernd)(/&%)(/&%Martin 

Der von folgenden Zeichen „befreit“ werden soll: 

%&/() 

 

Die Funktion WECHELN löscht das erste Zeichen: 

=WECHSELN(B2;LINKS(C2;1);"") 

 

Das erste Zeichen wird „entnommen“: 

=RECHTS(C2;LÄNGE(C2)-1) 

 

Das Ganze kann man herunterkopieren: 

 

Die beiden Formeln lauten also: 

=WECHSELN(B2;LINKS(C2;1);"") und =RECHTS(C2;LÄNGE(C2)-1) 

Ersetzt man die Zellnamen durch Parameternamen können sie wie folgt lauten: 

WECHSELN(Eingabetext;LINKS(Zeichen;1);"") und =RECHTS(Zeichen;LÄNGE(Zeichen)-1) 

Hinweis 

Die Verwendung des Begriffs „Zeichen“ ist nicht glücklich: ZEICHEN ist eine 
Funktion in Excel. ZEICHEN() wird als Funktion interpretiert, Zeichen ohne Klammer 
wird als Parameterwert interpretiert. Man sollte die Namen einheitlich wählen, 
beispielsweise mit einem Unterstrich beginnen: _Eingabezeichen und _Zeichen. So sind 
sie schnell zu finden und eindeutig. 

Diese Berechnungen werden so lange aufgerufen, bis Zeichen="". Das ist unser 
Abbruchkriterium: 

WENN(Zeichen="";Eingabetext;meinSÄUBERN(WECHSELN(Eingabetext;LINKS(Zeichen;1)
;"") und =RECHTS(Zeichen;LÄNGE(Zeichen)-1)) 

Und LAMBDA macht das möglich: 

=LAMBDA(Eingabetext;Zeichen;WENN(Zeichen="";Eingabetext;meinSÄUBERN(WECHSE
LN(Eingabetext;LINKS(Zeichen;1);"");RECHTS(Zeichen;LÄNGE(Zeichen)-1)))) 

Das Ergebnis ist eine Fehlermeldung #KALK 

Kopiert man allerdings diese Funktion in den Namensassistenten und speichert sie unter dem 
Namen meinSÄUBERN ab, kann man sie verwenden: 

 

Das Ergebnis: 

 

Oder so: 

5. Zwei Beispiele zum rekursiven Rechnen 

5.1. Die Schnecke 

Vor einer 4,50 Meter hohen Mauer sitzt eine Schnecke. Jeden Tag klettert diese Schnecke eine 
Höhe von 0,5 Meter hoch, jede Nacht rutscht sie 10 % der Gesamthöhe nach unten. Sie klettert 
also am ersten Tag von 0 Meter auf 0,5 Meter, um in der ersten Nacht um 10 % von 0,5 Meter, 
also um 0,05 Meter, hinunterzurutschen. Sie startet am nächsten Morgen auf einer Höhe von 0,45 
Meter. Erschöpft erreicht sie am Abend darauf 0,95 Meter. In der folgenden Nacht rutscht sie 
um 10 % von 0,95 Meter, also um 0,095 Meter, auf 0,85 Meter. Von dort klettert sie auf 1,355 
Meter, rutscht aber in der darauffolgenden Nacht auf 1,2195 Meter. Frage: Wie viele Tage muss 
die Schnecke klettern, wenn die Mauer eine Höhe von 4,5 Metern hat? 

Tag Morgenhöhe Abendhöhe 

Tag 1 0,00 m 0,50 m 

Tag 2 0,45 m 0,95 

Tag 3 0,855 m 1,355 m 

Tag 4 …  

 

Da die Strecke, welche die Schnecke nachts herunterrutscht, in jeder Nacht größer ist als in der 
vorhergehenden, kann diese Aufgabe nicht im Kopf gelöst werden. In der ersten Nacht hat die 
Schnecke eine Höhe von 0,5 Meter erreicht. In der zweiten Nacht befindet sie sich auf 0,5  

0,9+0,5 = 0,95. In der dritten Nacht beträgt ihre Höhe (0,5  0,9 + 0,5)  0,9 + 0,5 oder 0,5  (1 
+ 0,9 + 0,9²). Allgemein ist sie in der Nacht Nummer n auf 0,5  (1 + 0,9 + 0,9² +...+ 0,9n-1).  

Für die Summe 

 

𝑞 ൌ 𝑎 ൅ 𝑎𝑞 ൅ 𝑎𝑞ଶ ൅ ⋯ ൅ 𝑎𝑞௡ିଶ ൅ 𝑎𝑞௡ିଵ 

gilt: 

𝑠௡ ൌ 𝑎 ൈ
1 െ 𝑞௡

1 െ 𝑞
 

Da q = 0,9 und a = 0,5 sind, lautet die Formel: 

4,5 ൑ 0,5 ൈ
1 െ 0,9௡

1 െ 0,9
 

oder:  

0,9௡ ൑ 0,1 

Daraus folgt für n: 

𝑛 ൑ 𝑙𝑜𝑔଴,ଽ0,1 ൎ 21,854 

Also wird die Schnecke am 22. Tag die Mauer „bezwungen“ haben. 

Diese Aufgabe kann dagegen leicht in Excel gelöst werden. Eine Spalte dient zur Beschriftung 
der Tage, eine Zeile zur Beschriftung der Tageszeiten. In der ersten Spalte läuft ein Zähler. 

Da die Schnecke bei 0 startet, wird dies in die erste Zelle eingetragen. In der Abendhöhe steht 
nun die Formel Morgenhöhe + 0,5 oder: B2 + 0,5. 

 

Abbildung 5.1  Die Abendhöhe 

Werden nun von dieser Zahl 10 % abgezogen, so darf nicht C2-10 % geschrieben werden! Dies 
führt unweigerlich zu einem Fehler, da 10 % gleichbedeutend ist mit 0,1. Vielmehr muss 10 % 
von C2 abgezogen werden, also: 

=C2-C2*10% 

Oder einfacher: Da uns nicht die Strecke interessiert, die unsere Schnecke herunterrutscht (die 10 
%), sondern die Endhöhe (die 90 %), so kann diese auch ermittelt werden über: 

=C2*90% 

Man könnte die beiden Werte (10 % beziehungsweise 90 % und 0,5) auch auslagern, das heißt, in 
andere Zellen schreiben. Dann könnte man mit einem absoluten Bezug darauf zugreifen. Dies 
macht Sinn, wenn man alle möglichen Schnecken berechnen wollte.  

Diese Morgenhöhe wird mit einem Bezug in die nächste Zeile übernommen. 

Schließlich können alle drei Spalten heruntergezogen werden. 

Nun braucht man nur noch nachzusehen, wann die 4,5 Meter überschritten sind. Aber in der 
Abendspalte, nicht in einer der Morgenspalten! Denn die Schnecke klettert von 4 auf 4,5 Meter 
hinauf und rutscht nicht von 5 auf 4,5 Meter herunter. Übrigens: Wäre die Mauer 5,0 Meter hoch, 
so würde unsere arme Schnecke nie oben ankommen! 

 

Die drei Formeln lauten also 

=D2 und =B3+0,5 und =C3*0,9 

Speichert man sie als LAMBDA, könnte die Formel wie folgt aussehen: 

=LAMBDA(Tage;Morgenhöhe;Abendhöhe;wenn(Abendhöhe>=4,5;Tage;meineSchnecke(
Tage+1;Abendhöhe*0,9;Morgenhöhe+0,5))) 

Das Ergebnis von =meineSchnecke(1;0;0) liefert 44. Die korrekte Lösung dagegen lautet 22. Der 
Grund ist schnell gefunden: bei jedem Aufruf wird entweder die Morgenhöhe vergrößert und an 
die Abendhöhe übergeben oder umgekehrt die Abendhöhe verringert. Wir benötigen also nur die 
Hälfte der Schritte. Da die Abendhöhe von Interesse ist, muss man mit der Abendhöhe (also 0,5) 
beginnen: 

=LAMBDA(Tage;Abendhöhe;WENN(Abendhöhe>=4,5;Tage; 
meineSchnecke2(Tage+1;Abendhöhe*0,9+0,5))) 

Und =meineSchnecke2(1;0,5) liefert das korrekte Ergebnis 22. 

5.2. Beispiel 2: RMZ und KUMZINSZ 

Ein Schuldenbetrag von 10.000 € soll in 10 Jahren, also in 120 Rückzahlungen zurückbezahlt 
werden. Bei einem Jahreszinssatz von 2,5% Liegt die monatliche Annuität bei 94,27 € 

=RMZ(2,5%/12;120;-10000) 

Insgesamt wurden hierfür 

=KUMZINSZ(H3/12;120;H2;1;120;0) 

Also 1.312,39 €, was man durch Markieren der Summenspalte leicht zeigen kann. Dies soll 
berechnet werden. Beispielsweise so: 

=LAMBDA( 

Monat; 

Monatszähler; 

Anfangskapital; 

Annuität; 

Zinssatz; 

Zins; 

wenn(Monat<=Monatszähler;Zins;meineKUMZinsen( 

Monat; 

Monatszähler+1; 

Anfangskapital-Annuität+Anfangskapital*Zinssatz/12; 

Annuität; 

Zinssatz; 

Zins+Anfangskapital*Zinssatz/12)) 

Und so liefert 

 

=meineKUMZinsen(120;1;H2;H4;H3;0)  

den Wert 1.312,19 

6. WURDEAUSGELASSEN 

Eigentlich sollte die Funktion WURDEAUSGELASSEN innerhalb von LAMBDA funktioniert. 
Mit ihrer Hilfe kann auf fehlende Parameter reagiert werden. 

Michael Girvin zeigt in seinem Video  

https://www.youtube.com/watch?v=Eb5sXiLhUd4 

wie man mit LAMBDA die Funktion PercentChange erzeugen kann: 

=LAMBDA(Start;End;End/Start-1) 

Allerdings wird die englische Funktion ISOMMITTED ins Deutsche übersetzt mit 
WURDEAUSGELASSEN und funktioniert hier nicht mehr: 

=LAMBDA(Start;End;WENNS(UND(WURDEAUSGELASSEN(Start);WURDEAUSGELASSEN(End));
"Enter Both Start and End Amounts";WURDEAUSGELASSEN(Start);"Please Enter 
Start";WURDEAUSGELASSEN(End);"Please Enter End";WAHR;End/Start-1))(I4;J4) 

7. Weitere Funktionen, die LAMBDA verwenden: MAP 

Die Funktion MAP gibt ein Array zurück, das durch Zuordnung jedes Werts in den Arrays zu 
einem neuen Wert gebildet wird, indem eine LAMBDA angewendet wird, um einen neuen Wert 
zu erstellen. 

So kann beispielsweise berechnet werden: 

=MAP(B1:Q1;Ostersonntag) 

 

8. NACHSPALTE und NACHZEILE 

Analog arbeiten die beiden Funktionen NACHSPALTE und NACHZEILE. Sie liefern einen 
Bereich. Beispielsweise: 

=NACHSPALTE(D3:L53;LAMBDA(Spalte;MITTELWERT(Spalte))) 

Oder auch: 

=NACHZEILE(B2:AA13;LAMBDA(Bereich;MEDIAN(Bereich))) 

Da sich der Bereich nicht dynamisch vergrößert, bietet sich eine intelligente Tabelle an: 

=NACHZEILE(tbl_Schularbeit;LAMBDA(Bereich;MEDIAN(Bereich))) 

Hinweis 

Leider wird kein Ergebnis im Funktionsassistenten angezeigt: 

9. SCAN 

Erinnerst du dich an Reihen aus der Schulmathematik? Es geht um Summen von Zahlenfolgen, 
die einer Gesetzmäßigkeit genügen. Berechnet man beispielsweise  

1 ൅  
1
2

൅  
1
4

൅  
1
8

൅
1

16
൅ ⋯ 

Also: 

෍
1

𝑛ଶ

ஶ

௡ୀଵ

 

So konvergiert diese Reihe gegen 2. Über einen endlichen Bereich kann die Funktion SCAN dies 
für jeden Wert berechnen: 

=SCAN(0;A1:A32;LAMBDA(Ergebnis;Wert;Ergebnis+1/(Wert^2))) 

Oder auch  

=SCAN(0;A1:A32;LAMBDA(a;b;a+1/FAKULTÄT(b-1))) 

Welche gegen die Eulersche Zahl e konvergiert. 

Mit dieser Hilfe kann man die ISBN-13 berechnen: 

=REST(SUMME(SCAN(0;I3:I11;LAMBDA(a;b;1+a))*(I3:I11));11) 

10. REDUCE 

Einen ähnlichen Weg geht REDUCE. Auch diese Funktion greift auf einen Bereich zu, 
berechnet jedoch nicht ein Ergebnis für jeden einzelnen Wert, sondern für liefert ein Ergebnis. 
Das heißt: im oberen Beispiel hätte man REST und SUMME durch REDUCE ersetzen können. 

Oder: summiere die geraden Werte: 

=REDUCE(0;A1:A13;LAMBDA(a;b;WENN(ISTGERADE(b);a+b;a))) 

Die Koordinaten von München (beispielsweise Marienplatz) sind  

Lat: 48,1371079 und Lon: 11,5753822. 

Die vom Roten Platz in Moskau lauten 

Lat: 55,7536283 und Lon: 37,6213796006738 

Das kann man beispielsweise über 

https://www.koordinatengps.de/ 

herausfinden. Die Entfernung zweier Punkte kann man nicht mit dem Satz des Pythagoras 
berechnen, sondern mit Hilfe von sphärischer Trigonometrie. Ein Blick in die 
Formelsammlung oder ins Internet liefert die Lösung: 

Entfernung = 6378,388 * acos(sin(lat1) * sin(lat2) + cos(lat1) * cos(lat2) 
* cos(lon2 - lon1)) 

Da Sinus und Cosiuns von einer Einheitskugel ausgehen, muss das Ergebnis mit dem Radius 
der Erde (ungefähr 6.380 km) multipliziert werden. Und da Excel mit der Funktion 
BOGENMASS diese Angaben in GRAD umrechnet, lautet die Formel: 

= 6378*ARCCOS(SIN(BOGENMASS(B2))*SIN(BOGENMASS(B3))+ 
COS(BOGENMASS(B2))*COS(BOGENMASS(B3))*COS(BOGENMASS(C3-C2))) 

 

 

Das kann man doch sicherlich mit den neuen Arrayfunktionen, beispielsweise mit LAMBDA 
und REDUCE abkürzen. Da zwei Mal der COSINUS verwendet wird und ein drittes Mal der 
Cosinus einer Differenz, ermittle ich die Differenz unterhalb der Daten: 

 

Und berechne nun: 

= 6378*ARCCOS(REDUCE(1;B2:B3;LAMBDA(a;b;a*SIN(BOGENMASS(b))))+ 
REDUCE(1;B2:B4;LAMBDA(a;b;a*COS(BOGENMASS(b))))) 

Die Formel ist etwas kürzer als die erste: 

 

Okay – München – Moskau sind 1.962 km Luftlinie. Und nach Madrid? Ich hole die Daten 
aus: 

https://www.koordinatengps.de/ 

Nach Madrid sind es von München aus „nur“ 1.486 km Luftlinie – ist also näher als Moskau. 

11. MATRIXERSTELLEN 

Während die Funktion REDUCE eine Liste von Berechnungen auf einen Wert reduziert, baut 
MATRIXERSTELLEN eine Matrix auf. Mit Hilfe dieser Funktion kann man den BMI 
berechnen, der sich als Gewicht / (Größe x Größe) berechnet: 

=MATRIXERSTELLEN(100;120;LAMBDA(Gewicht;Größe; 
(Gewicht+49)/(((100+Größe-1)/100)^2))) 

Mit einer bedingten Formatierung kann man da Ganze gut visualisieren. 

 

Damit kann das Problem von Euler visualisiert werden: Er behauptete, dass zwei verschiedene 
ganze Zahlen n und m nur mit 2 und 4 folgende Gleichung lösen: 

nm = mn 

Dazu gibt man in die erste Zelle die Formel 

=MATRIXERSTELLEN(100;100;LAMBDA(m;n;m^n=n^m)) 

ein, Die Zellen, die WAHR liefern, werden farblich mit Hilfe der bedingten Formatierung 
gekennzeichnet. 

12. MATRIXZUTEXT 

Der Sinn der Funktion MATRIXZUTEXT erschließt sich mir noch nicht ganz. Die Funktion 

=MATRIXZUTEXT(A9:B14) 

Berechnet das Gleiche wie 

=TEXTVERKETTEN("; ";WAHR;A9:B14) 

Lediglich mit dem Parameterwert 1 (streng) bei „Format“ wird ein Unterschied deutlich: 

=MATRIXZUTEXT(A9:B14;1) 

13. Fazit 

Die neuen Funktion LAMBDA & co erweitern das Spektrum von Excel. Wurden mit den 
Arrayfunktionen SORTIEREN, FILTERN, SEQUENZ, EINDEUTIG & co im ersten Schritt 
eine Erweiterung der „alten“ {Matrixfunktionen} geschaffen, so ist es jetzt möglich, eigene 
Array- (oder Spill-)funktionen zu schreiben. Die sechs Funktionen MAP, NACHSPALTE, 
NACHZEILE, SCAN, REDUCE und MATRIXERSTELLEN liefern sämtliche Varianten: von 
Liste zu Liste; von Liste zu Wert, von Wert zu Liste, von Liste zu Spalte/Zeile oder auch die 
Einzelberechnung der Werte der Liste. 

Der Aufbau der LAMBDA-Funktionen ist sicherlich gewöhnungsbedürftig; bedauerlich ist auch 
der fehlende Einblick in die „Maschine“ – eine schrittweise Kontrolle der Berechnungen ist nicht 
möglich, auch fehlen noch einige Hilfen beim Editieren und Verändern der Funktionen. 

Dennoch: mit Einführung dieser Funktionen hat Microsoft sicherlich eine weitere, interessante 
Technologie eingeführt, mit deren Hilfe einige Lösungen schneller und einfacher erstellt werden 
können. Welche – das wird sicherlich die Zukunft zeigen. 


